

## **Combining an Active Technologies Dashboard With an HTML Page Using -HTMLFORM**

Using the WebFOCUS Dialogue Manager command -HTMLFORM, you can combine an active dashboard with an HTML page to create a single HTML document. You can provide links between the active dashboard and the HTML page to enable end user navigation within the document. The HTML page can contain an HTML form, HTML dashboard, or any other similar HTML-supported item.

This Technical Memo provides a sample procedure (FEX) that combines an active dashboard with an HTML page to create a single HTML document, such as an interactive e-statement.

### **Overview**

On a high level, the sample procedure does the following:

- ❑ Sets the Active Technologies driver in the WebFOCUS Reporting Server to generate styling structures and data structures only.
- ❑ Creates a HOLD file for the active dashboard.
- ❑ Specifies the icon set and style set used in the active dashboard, and resets the Active Technologies driver in the WebFOCUS Reporting Server to generate the JavaScript.
- ❑ Creates an inline HTML form in the procedure. Within the inline HTML form, there is code that does the following:
  - ❑ Provides information about the HTML document.
  - ❑ Embeds Active Technologies JavaScript in the HTML output.
  - ❑ Calls JavaScript functions that show the active dashboard and the HTML page. It also sets the properties of the container for the active dashboard and the HTML page. The HTML page displays a short text message.
  - ❑ Calls the HOLD file for the active dashboard and specifies where to display the active dashboard.

- ❑ Creates the links for navigation between the active dashboard and the HTML page.

## Creating the Procedure

The sample procedure follows. It is named **combination.fex**.

**Note:** An explanation of a line of code or block of code precedes the line or block.

For each SET command in the procedure, the explanation above it indicates where you can find more information on that command. You can also find more information on:

- ❑ COMPOUND syntax in the *Active Technologies User's Guide* and the *Creating Reports With WebFOCUS Language* manual. In the *Creating Reports With WebFOCUS Language* manual, see in particular the topic *Creating a Compound Report* in Chapter 10, *Linking a Report to Other Resources*.
- ❑ WebFOCUS syntax in the *Creating Reports With WebFOCUS Language* manual.
- ❑ Dialogue Manager syntax in the *Developing Reporting Applications* manual.

### **combination.fex**

The first command in the procedure, SET AROUTPUT, is WebFOCUS syntax that controls what is generated as the Active Technologies output file in the WebFOCUS Reporting Server. In this example, it is set to the value DATALOOK, which generates styling structures and data structures only. The output produced is not a complete HTML active report. For more information on the SET AROUTPUT command, see *How to Control the Generation of the Active Technologies for HTML Output File*, in the topic *Creating an Active Technologies Report*, in Chapter 10, *Creating Active Technologies Components With WebFOCUS Syntax*, of the *Active Technologies User's Guide*.

```
SET AROUTPUT=DATALOOK
```

The COMPOUND LAYOUT command that follows is WebFOCUS syntax that begins the code that defines and generates an active dashboard. Use HOLD in the command to create a reusable file that contains the output for the active dashboard. In this example, the HOLD file is named XYZ, and its format is AHTML. The COMPOUND END command, later in the procedure, signifies the end of the compound report. In this example, the active dashboard contains two page layouts. The first page displays a tabular active report and an active bar chart. The second page displays another tabular active report.

```
COMPOUND LAYOUT HOLD AS XYZ FORMAT AHTML
```

*Combining an Active Technologies Dashboard With an HTML Page Using -HTMLFORM*

```
UNITS=IN, $
SECTION=section1, LAYOUT=ON, METADATA='0.5^0.5^0.5^0.5', MERGE=AUTO,
  ORIENTATION=PORTRAIT, PAGESIZE=Letter, ARGRAPHENGINE=JSCHART, $
PAGELAYOUT=1, NAME='Page layout 1', text='Page layout 1', TOC-LEVEL=1,
  BOTTOMMARGIN=0.5, TOPMARGIN=0.5, METADATA='BOTTOMMARGIN=0.5,
  TOPMARGIN=0.5,LEFTMARGIN=0,RIGHTMARGIN=0,', $
COMPONENT='report1', TEXT='report1', TOC-LEVEL=2, POSITION=(0.667 0.719),
  DIMENSION=(3.813 3.021), METADATA='Z-INDEX: 100; LEFT: 0.667in;
  OVERFLOW: auto; WIDTH: 3.813in; POSITION: absolute; TOP: 0.719in;
  HEIGHT: 3.021in', $
COMPONENT='report2', TEXT='report2', TOC-LEVEL=2, POSITION=(5.719 0.708),
  DIMENSION=(5.698 5.167), METADATA='Z-INDEX: 100; LEFT: 5.719in;
  OVERFLOW: auto; WIDTH: 5.698in; POSITION: absolute; TOP: 0.708in;
  HEIGHT: 5.167in', $
PAGELAYOUT=2, NAME='Page layout 2', text='Page layout 2', TOC-LEVEL=1,
  BOTTOMMARGIN=0.5, TOPMARGIN=0.5, METADATA='BOTTOMMARGIN=0.5,
  TOPMARGIN=0.5,LEFTMARGIN=0,RIGHTMARGIN=0,', $
COMPONENT='report3', TEXT='report3', TOC-LEVEL=2, POSITION=(0.667 0.719),
  DIMENSION=(3.813 3.021), METADATA='Z-INDEX: 100; LEFT: 0.667in;
  OVERFLOW: auto; WIDTH: 3.813in; POSITION: absolute; TOP: 0.719in;
  HEIGHT: 3.021in', $
END

SET COMPONENT='report1'
-*component_type report
TABLE FILE GGSales
SUM
DOLLARS
BUDDOLLARS
BY REGION
BY ST
ON TABLE SET PAGE-NUM OFF
ON TABLE NOTOTAL
ON TABLE PCHOLD FORMAT AHTML
ON TABLE SET HTMLCSS ON
ON TABLE SET STYLE *
INCLUDE=ENInformationBuilders_Medium1.sty,$
ENDSTYLE
END
```

## Creating the Procedure

```
SET COMPONENT='report2'
-*component_type report
GRAPH FILE GGSales
SUM
UNITS
BUDUNITS
BY REGION
BY ST
BY CITY
ON TABLE PCHOLD FORMAT AHTML
ON GRAPH SET HTMLENCODE ON
ON GRAPH SET GRAPHDEFAULT OFF
ON GRAPH SET ARGGRAPHENGIN JSCHART
ON GRAPH SET VZERO OFF
ON GRAPH SET LOOKGRAPH VBAR
ON GRAPH SET STYLE *
INCLUDE=ENInformationBuilders_Medium1.sty,$
TYPE=HEADING,SIZE=12,JUSTIFY=CENTER,COLOR=RGB(#484e57),$
TYPE=REPORT,
    COLUMN=N1,
    HIDE=ON,
$
TYPE=REPORT,
    COLUMN=N2,
    HIDE=ON,
$
ENDSTYLE
END

SET COMPONENT='report3'
-*component_type report
TABLE FILE GGSales
SUM
UNITS
BUDUNITS
BY REGION
BY ST
BY CITY
ON TABLE SET PAGE-NUM OFF
ON TABLE NOTOTAL
ON TABLE PCHOLD FORMAT AHTML
ON TABLE SET HTMLCSS ON
ON TABLE SET STYLE *
INCLUDE=ENInformationBuilders_Medium1.sty,$
ENDSTYLE
END
COMPOUND END
-RUN
```

The code for the compound report is followed by three SET commands. The SET ARICONSET command is WebFOCUS syntax that specifies the icon set used in the active dashboard. In this example, the icon set is supplied with the product and is named *white*. For more information on the SET ARICONSET command, see *How to Customize Icons for an Active Technologies Report or Dashboard*, in the topic *Creating an Active Technologies Report*, in Chapter 10, *Creating Active Technologies Components With WebFOCUS Syntax*, of the *Active Technologies User's Guide*. Also see *Appendix B, Creating a Custom Icon Set for an Active Technologies Report or Dashboard*, of the *Active Technologies User's Guide*.

```
SET ARICONSET=white
```

The SET ARSTYLESET command specifies the style set used in the active dashboard. In this example, the style set is supplied with the product and is named *bipib*. For more information on the SET ARSTYLESET command, see *How to Apply Custom Styles to an Active Technologies Report or Dashboard*, in the topic *Creating an Active Technologies Report*, in Chapter 10, *Creating Active Technologies Components With WebFOCUS Syntax*, of the *Active Technologies User's Guide*. Also see *Appendix C, Defining Custom Styles for an Active Technologies Report or Dashboard*, of the *Active Technologies User's Guide*.

```
SET ARSTYLESET=bipib
```

The SET AROUTPUT command controls what is generated as the Active Technologies output file in the WebFOCUS Reporting Server. Use the value ALL to reset the Active Technologies driver and allow the generation of Active Technologies JavaScript. For more information on the SET AROUTPUT command, see *How to Control the Generation of the Active Technologies for HTML Output File*, in the topic *Creating an Active Technologies Report*, in Chapter 10, *Creating Active Technologies Components With WebFOCUS Syntax*, of the *Active Technologies User's Guide*.

```
SET AROUTPUT=ALL
```

Following the three SET commands is the -HTMLFORM BEGIN command, which is standard Dialogue Manager -HTMLFORM syntax. -HTMLFORM BEGIN indicates the beginning of an inline HTML form in a procedure.

```
-HTMLFORM BEGIN
```

The following customized block of HTML code includes HTML elements such as <HEAD> and <META>. Use standard HTML here to provide information about the HTML document.

```
<!DOCTYPE html>
<HTML>
<HEAD>
<META name="HandheldFriendly" content="True">
<META name="PalmComputingPlatform" content="True">
<TITLE>Active Technologies Dashboard</TITLE>
</HEAD>
<BODY>
```

The `!IBI.OBJ.ACTIVEREPORTJS;` command is WebFOCUS syntax that embeds Active Technologies JavaScript in the HTML output.

```
!IBI.OBJ.ACTIVEREPORTJS;
```

The next customized block of HTML code includes two HTML `<a href>` elements. In this example, the `<a href>` element is used to call customized JavaScript functions from the script section later in this procedure. The functions are named `showReport` and `showForm`. These elements are also used to specify the text that labels the links above the active dashboard and HTML page. In this example, the text is *report* and *form*. Even though the links are not customized in this example, they can be customized to look like buttons or tabs using standard cascading style sheets (CSS) if needed.

```
<a href="javascript:showReport();">report</a>
<a href="javascript:showForm();">form</a>
```

The following customized block of HTML code includes two `<div>` elements. In this example, the `<div>` element is used to set the ID of the container in which the active dashboard is displayed and the ID of the container in which the HTML page is displayed. The background color of each container is also set in the `<div>` element. The container ID is the `<div id>` *PutReport* and the `<div id>` *PutForm*. In this example, the following standard HTML also displays the text message *Enter your html form here!* on the HTML page.

```
<div style="position:relative;width:1024px;height:768px">
  <div id="PutReport" style="position:absolute;z-index:1;x:0px;
    y:0px;background-color:#B0E0E6;width:1024px;height:768px">
  </div>
  <div id="PutForm" style="position:absolute;z-index:0;x:0px;
    y:0px;background-color:yellow;width:1024px;height:768px">
    Enter your html form here!
  </div>
</div>

<SCRIPT LANGUAGE="JavaScript">
<!--
```

The `!IBI.FIL.XYZ;` command is WebFOCUS syntax that calls a file. In this example, it calls the `HOLD` file for the active dashboard named `XYZ`, which you generated earlier in this procedure.

```
!IBI.FIL.XYZ;
```

The next line of code calls `genTables`, which is an Active Technologies JavaScript function. It takes two parameters. The first parameter is an internal true or false flag. The second parameter is a container ID. The value *true* for the flag is used internally only. Always use *false* for a customized HTML form, as in this example. Include the container ID *PutReport* from the previous block of HTML code. *PutReport* indicates where to display the active dashboard.

```
genTables(false,"PutReport");
```

You can include customized JavaScript after you call the Active Technologies JavaScript. In this example, the customized JavaScript creates the HTML links labeled *report* and *form* above the active dashboard and HTML page and provides navigation between the two pages.


```
function showReport(){
  document.getElementById("PutReport").style.zIndex=1;
  document.getElementById("PutForm").style.zIndex=0;
}
function showForm(){
  document.getElementById("PutReport").style.zIndex=0;
  document.getElementById("PutForm").style.zIndex=1;
}
//-->
</SCRIPT>
</BODY>
</HTML>
```

The code terminates with the -HTMLFORM END command, which is standard Dialogue Manager -HTMLFORM syntax. -HTMLFORM END indicates the end of the inline HTML form in the procedure.

**-HTMLFORM END**

## Running the Procedure

When you run the procedure, the following page of the active dashboard is displayed. The cursor is pointing to the link that is labeled *report*. The report link accesses this active dashboard page.



You can use standard HTML and CSS to style the links above the active dashboard and HTML page as desired. For example, you can set the style so that the links look like buttons.

The background color #B0E0E6, the container width 1024px, and the height 768px are set in the div container named *PutReport*.

```
<div style="position:relative;width:1024px;height:768px">  
  <div id="PutReport" style="position:absolute;z-index:1;x:0px;  
    y:0px;background-color:#B0E0E6;width:1024px;height:768px">  
  </div>
```

- .
- .
- .




## Combining an Active Technologies Dashboard With an HTML Page Using -HTMLFORM

Both page layouts of the compound report are part of the report link. They use the white icon set and the bipib style set.

```
SET ARICONSET=white
SET ARSTYLESET=bipib
```

When you click the *form* link, the following HTML page is displayed. In this example, it is blank except for the text message *Enter your html form here!*



The background color yellow, the container width 1024px, the height 768px, and the text message are set in the div container named *PutForm*.

```
<div style="position:relative;width:1024px;height:768px">
.
.
.
  <div id="PutForm" style="position:absolute;z-index:0;x:0px;
    y:0px;background-color:yellow;width:1024px;height:768px">
    Enter your html form here!
  </div>
</div>
```

*Running the Procedure*